```
/*****************************************************************************************************;
 *   COMPANY                       : TAKEDA
 *   PROTOCOL/STUDY/PROJECT        : ShortCourse
 *   AUTHOR                        : OS Adjusted Analysis Working Group
 *   DATE                          : 08Sep2025
 *   PROGRAM NAME                  : case_study_1way_MSM_clean.sas
 *   LOCATION                      : /.../.../...
 *   VERSION/ PLATFORM             : SAS V9.4 / LINUX
 *   PURPOSE                       : Program to Perform Marginal Structural Model (MSM) Analysis - 1 Way Crossover.
 *   MACROS CALLED                 :
 *   INPUT                         :
 *   OUTPUT                        :
 *   PARAMETERS (Optional)         :
 *                                 :
 *   NOTES (Optional)              : ASA Traveling Course, September to October 2025
 *
 *****************************************************************************************************;
 *****************************************************************************************************/
options noquotelenmax;

** define relevant directories ;
libname ads "/bdm/tbos/AP24534/studies/201/csr/adhoc/ShortCourse/Clean_SAS_Programs" access=readonly;
%include "/bdm/tbos/AP24534/studies/201/csr/adhoc/ShortCourse/Demo/macro_km_plot.sas";

data final;
      set ads.final_casestudy2;
      by usubjid analdy;
      month2=month*month;
      month3=month2*month;
      drop altrx althdy;

run;

*** trt01pn=2 is active;

data finala;
      set final;
      by usubjid analdy;

      if trt01pn=2;
run;
```

```
*** trt01pn=1 is placebo;

data finalb;
      set final;
      by usubjid analdy;

      if trt01pn=1;
run;

ods exclude all;
***********************************************************;
*** creating weight
**************************************************************;

/* Model 1, weight for taking alternativ therapy, numerator */
/* stratification factors forced in, for other covariates, p-value is small in at least one models */
proc logistic data=finalb;
      where analdy<=xoverdy or xoverdy=.;
      class trt01p lyticbone_model indctreg asctresp mrd_model time_induct_trt
            cytocat CRECEL_finalmodel;
      model xover=ASCTRESP CYTOCAT MONTH baseHGB_model baseLDH_model basePLAT_model
            crecel_finalmodel duraEx_model extramed_model meltype_model month2 month3;
      output out=model1a p=altv_0;
run;
/*Assign IPTW=1 for active arm cohort*/
data model1;
      set model1a finala(in=infinala);
      if infinala then altv_0=1;
run;

proc sort;
      by usubjid analdy;
run;

/* Model 2, weight for taking alternativ therapy, denominator */
proc logistic data=finalb;
      where analdy<=xoverdy or xoverdy=.;
      class trt01p indctreg asctresp lyticbone_model mrd_model time_induct_trt
            cytocat CRECEL_finalmodel;
      model xover=ASCTRESP CYTOCAT LOCFHGB LOCFPLAT LOCFSPEP MONTH PD baseHGB_model
            baseLDH_model basePLAT_model crecel_finalmodel duraEx_model extramed_model
            meltype_model month2 month3;
```

```
      output out=model2a p=altv_w;
run;

/*Assign IPTW=1 for active arm cohort*/
data model2;
      set model2a finala(in=infinala);
      if infinala then altv_w=1;
run;

proc sort;
      by usubjid analdy;
run;

/* Model 3 , weight for non-informative censoring, numerator */
proc logistic data=final;
      class trt01p indctreg asctresp lyticbone_model mrd_model time_induct_trt
            cytocat CRECEL_finalmodel;
      model censor=ASCTRESP CYTOCAT MONTH baseHGB_model baseLDH_model basePLAT_model
            crecel_finalmodel duraEx_model extramed_model meltype_model month2 month3;
      output out=model3 p=punc_0;
run;

/* Model 4  weight for non-informative censoring, denominator */
proc logistic data=final;
      class trt01p indctreg asctresp lyticbone_model mrd_model time_induct_trt
            cytocat CRECEL_finalmodel;
      model censor=ASCTRESP CYTOCAT LOCFHGB LOCFPLAT LOCFSPEP MONTH PD baseHGB_model
            baseLDH_model basePLAT_model crecel_finalmodel duraEx_model extramed_model
            meltype_model month2 month3;
      output out=model4 p=punc_w;
run;

data main_w;
      merge model1 model2 model3 model4;
      by usubjid analdy;

      /* variables ending with _0 refer to the numerator of the weights
      Variables ending with _w refer to the denominator of the weights */
      *** compute stablized and non-stablized weights, stablized weight will be used;

      /* reset the variables for a new patient */
      if first.usubjid then
```

```
            do;
                  k1_0=1;
                  k2_0=1;
                  k1_w=1;
                  k2_w=1;
            end;
      retain k1_0 k2_0 k1_w k2_w;

      /* Inverse probability of censoring weights */
      k2_0=k2_0*punc_0;
      k2_w=k2_w*punc_w;

      /* Inverse probability of treatment weights */
      /* patients not on alternative therapy */
      if analdy<xoverdy or xoverdy=. then
            do;
                  k1_0=k1_0*altv_0;
                  k1_w=k1_w*altv_w;
            end;

      /* patients that start alternative therapy in the current month */
      else if analdy=xoverdy then
            do;
                  k1_0=k1_0*(1-altv_0);
                  k1_w=k1_w*(1-altv_w);
            end;

      /* patients that have already started alternative therapy */
      else
            do;
                  k1_0=k1_0;
                  k1_w=k1_w;
            end;

      /* Stabilized and non stabilized weights */
      stabw=(k1_0*k2_0)/(k1_w*k2_w);
      nstabw=1/(k1_w*k2_w);
run;

proc sort data=main_w;
      by usubjid analdy;
run;
```

```
*** model check;

proc sort data=main_w out=main_wa;
      by trt01p;
run;

proc univariate data=main_wa noprint;
      var stabw;
      by trt01p;
      output out=check1 n=n mean=mean median=median std=std min=min max=max p95=p95
            p99=p99 p1=p1;
run;

ods exclude all;

proc print data=check1;
      title "check weight: mean should be near 1, MSM weight";
      title2 "Weight distribution is fine, no extreme large outliers";
      title3 "Accomodating possible informative censoring in both arms";
      var trt01p n mean std min max;
run;

data wtplot;
      set main_wa;
      q=floor(month/3);

      if q>14 then
            q=14;

      if stabw ne .;
      keep usubjid q trt stabw;
run;

proc univariate data=wtplot noprint;
      var q stabw;
      output out=tp n=nq ns mean=meanq means stderr=seqq ses;
run;

*** rescale stabw if there exist big outlier weights;

data main_w1a;
```

```
      set main_w;
      by usubjid analdy;
      mid=1;
run;

proc sort;
      by trt01p mid;

data check1;
      set check1;
      by trt01p;
      mid=1;
      keep trt01p p99 mid;
run;

data main_w1b;
      merge main_w1a check1;
      by trt01p mid;
      q=floor(month/3);

      if stabw>p99*2 then
            stabw=min(p99*1.3*(1+ranuni(analdy)), stabw);
run;

proc univariate data=main_w1b noprint;
      var stabw;
      by trt01p;
      output out=check2 n=n mean=mean median=median std=std min=min max=max p99=p99
            p1=p1;
run;

proc print data=check2;
      title "recheck weight: mean should be near 1, MSM weight";
      var trt01p n mean std min max;
run;

proc sort data=main_w1b out=main_w;
      by usubjid analdy;
run;

data main_w2;
      set main_w;
```

```
      by usubjid analdy;
      *** data error;
      a=lag(analdy);

      if first.usubjid then
            do;
                  start=0;
                  stop=analdy;
            end;
      else
            do;
                  start=a;
                  stop=analdy;
            end;

      if stop=start then
            stop=stop+1;
run;

data main_itt;
      set main_w;
      by usubjid analdy;

      if last.usubjid;
run;

ods select ModelANOVA ParameterEstimates;
proc phreg data=main_itt;
      class trt01p(ref='Placebo') indctreg asctresp stgenrl;
      model analdy*die(0)=trt01p /rl;
      strata indctreg asctresp stgenrl;
      title "look at ITT, not using counting process data, result is biased before adjustment";
run;

*** final MSM model;
ods select ModelANOVA ParameterEstimates;
ods output ParameterEstimates=stat_msm(where=(parameter='TRT01P'));
proc phreg data=main_w2 covs(aggregate);
      class trt01p(ref='Placebo') indctreg asctresp stgenrl;
      model (start, stop)*die(0)=trt01p xover /rl;
      strata indctreg asctresp stgenrl;
      id usubjid;
```

```
      weight stabw;
      title "MSM adjusting confounding from Xover";
      title2 "Using proc phreg, counting process data format, must use robust variance";
run;

****************************************************************************;
*** Call macro to plot adjusted KM curves
****************************************************************************;
%macro_km_plot(inmain=main_w2               /* Final weighted analysis data */
            ,instratds=stat_msm             /* Final statistic data contains stratified p-value, HR and 95% CI */
            ,inunstratds=                   /* Final statistic data contains unstratified p-value */
            ,method=MSM                     /* Weighted method */
            ,xover=Y                        /* Xover Y/N */
            );
```