```
/****************************************************************************************************;
 *   COMPANY                            : TAKEDA
 *   PROTOCOL/STUDY/PROJECT             : ShortCourse
 *   AUTHOR                             : OS Adjusted Analysis Working Group
 *   DATE                               : 08Sep2025
 *   PROGRAM NAME                       : case_study_1way_IPCW_clean.sas
 *   LOCATION                           : /.../.../...
 *   VERSION/ PLATFORM                  : SAS V9.4 / LINUX
 *   PURPOSE                            : Program to Perform Inverse Probability of Censoring Weighting
 *                                         (IPCW) Analysis - 1 Way Crossover.
 *   MACROS CALLED                      :
 *   INPUT                              :
 *   OUTPUT                             :
 *   PARAMETERS (Optional)              :
 *                                      :
 *   NOTES (Optional)                   : ASA Traveling Course, September to October 2025
 *
 ****************************************************************************************************;
 ****************************************************************************************************/
options noquotelenmax;
** define relevant directories ;
libname ads "/bdm/tbos/AP24534/studies/201/csr/adhoc/ShortCourse/Clean_SAS_Programs" access=readonly;
%include "/bdm/tbos/AP24534/studies/201/csr/adhoc/ShortCourse/Demo/macro_adj_pvalue.sas";
%include "/bdm/tbos/AP24534/studies/201/csr/adhoc/ShortCourse/Demo/macro_km_plot.sas";

data final;
      set ads.final_casestudy2;
      by usubjid analdy;
      retain censor2;

      if first.usubjid then censor2=0;
      month2=month*month;
      month3=month2*month;
      *** For IPCW, both altrx and censor into account;

      if (censor=1 or xover=1) then censor2=1;
run;


ods exclude all;
**********************************************************;
*** creating weight
```

```
**************************************************************;

/* final Model 1, weight for non-informative censoring, numerator */
proc logistic data=final;
      class trt01p lyticbone_model indctreg asctresp mrd_model time_induct_trt
            cytocat CRECEL_finalmodel;
      model censor2=ASCTRESP CYTOCAT MONTH TRT01P baseHGB_model baseLDH_model
            basePLAT_model crecel_finalmodel duraEx_model ecog_model meltype_model month2
            month3;
      output out=model1 p=altv_0;
run;

proc sort;
      by usubjid analdy;
run;

/* final Model 2 , weight for non-informative censoring, denominator */
proc logistic data=final;
      class trt01p indctreg asctresp lyticbone_model mrd_model time_induct_trt
            cytocat CRECEL_finalmodel;
      model censor2=ASCTRESP CYTOCAT LOCFHGB LOCFPLAT MONTH PD TRT01P baseHGB_model
            baseLDH_model basePLAT_model crecel_finalmodel duraEx_model ecog_model
            meltype_model month2 month3;
      output out=model2 p=altv_w;
run;

data main_w;
      merge model1 model2;
      by usubjid analdy;

      /* variables ending with _0 refer to the numerator of the weights
      Variables ending with _w refer to the denominator of the weights */
      if first.usubjid then
            do;
                  k1_0=1;
                  k1_w=1;
            end;
      retain k1_0 k1_w;

      /* Inverse probability of censoring weights, they are IPTW in thw MSM model */
      if analdy<xoverdy or xoverdy=. then
            do;
```

```
                    if ^missing(altv_0) then k1_0=k1_0*altv_0;
                         else k1_0=.;
                    if ^missing(altv_w) then k1_w=k1_w*altv_w;
                         else k1_w=.;
             end;

      /* patients that start alternative therapy in the current month */
      else if analdy=>xoverdy then
             do;
                   k1_0=0;
                   k1_w=k1_w;
             end;

      /* Stabilized weights only */
      if not(missing(k1_0) or missing(k1_w)) then stabw=k1_0/k1_w;

      if stabw=0 then
             delete;
run;

proc sort data=main_w;
      by usubjid analdy;
run;

*** model check;

proc sort data=main_w out=main_wa;
      by trt01p;
run;

proc univariate data=main_wa noprint;
      var stabw;
      by trt01p;
      output out=check1 n=n mean=mean median=median std=std min=min max=max p95=p95
             p99=p99 p1=p1;
run;

proc print data=check1;
      title "check weight: mean should be near 1, IPCW weight";
      title2 "Weight distribution is fine, no extreme large outliers";
      title3 "Accomodating possible informative censoring in both arms";
      var trt01p n mean std min max;
```

```
run;

data wtplot;
      set main_wa;
      q=floor(month/3);

      if q>14 then
            q=14;

      if stabw ne .;
      keep usubjid q trt stabw;
run;

proc univariate data=wtplot noprint;
      var q stabw;
      output out=tp n=nq ns mean=meanq means stderr=seqq ses;
run;

*** rescale stabw if there exist big outlier weights;

data main_w1a;
      set main_w;
      by usubjid analdy;
      mid=1;
run;

proc sort;
      by trt01p mid;

data check1;
      set check1;
      by trt01p;
      mid=1;
      keep trt01p p99 mid;
run;

data main_w1b;
      merge main_w1a check1;
      by trt01p mid;
      q=floor(month/3);

      if stabw>p99*2 then
```

```
            stabw=min(p99*1.3*(1+ranuni(analdy)), stabw);
run;

proc univariate data=main_w1b noprint;
      var stabw;
      by trt01p;
      output out=check2 n=n mean=mean median=median std=std min=min max=max p99=p99
            p1=p1;
run;

proc print data=check2;
      title "recheck weight: mean should be near 1, IPCW weight";
      var trt01p n mean std min max;
run;

proc sort data=main_w1b out=main_w;
      by usubjid analdy;
run;

*** reorganize data for proc phreg;

data main_w2b;
      set main_w;
      by usubjid analdy;
      *** data error;
      a=lag(analdy);

      if first.usubjid then
            do;
                  start=0;
                  stop=analdy;
            end;
      else
            do;
                  start=a;
                  stop=analdy;
            end;

      if stabw=. then
            stabw=1;
run;
```

```
*** final IPCW model;

ods select ModelANOVA ParameterEstimates;
ods output ParameterEstimates=stat_ipcw(where=(parameter='TRT01P'));
proc phreg data=main_w2b covs(aggregate);
      class trt01p(ref='Placebo') indctreg asctresp stgenrl;
      model (start, stop)*die(0)=trt01p /rl;
      strata indctreg asctresp stgenrl;
      id usubjid;
      weight stabw;
      title "IPCW Cox model for OS";
      title2 "Using proc phreg, counting process data format for multiple records";
      title3 "Accomodating possible informative censoring in both arms";
run;

*******************************************************************************;
*** Call macro to calcuate z-score and p-value from weighted Log-rank test
*******************************************************************************;
%macro_adj_pvalue(inds=main_w2b       /* Final weighted analysis data */
                  ,method=IPCW        /* Weighted method */
                  );

*******************************************************************************;
*** Call macro to plot adjusted KM curves
*******************************************************************************;
%macro_km_plot(inmain=main_w2b              /* Final weighted analysis data */
            ,instratds=stat_ipcw            /* Final statistic data contains stratified p-value, HR and 95% CI */
            ,inunstratds=stat_uspv_ipcw     /* Final statistic data contains unstratified p-value */
            ,method=IPCW                    /* Weighted method */
            ,xover=Y                        /* Xover Y/N */
            );
```