```
/*************************************************************************************************;
*   COMPANY                     : TAKEDA
*   PROTOCOL/STUDY/PROJECT      : ShortCourse
*   AUTHOR                      : OS Adjusted Analysis Working Group
*   DATE                        : 08Sep2025
*   PROGRAM NAME                : case_study_2way_MSM_clean.sas
*   LOCATION                    : /.../.../...
*   VERSION/ PLATFORM           : SAS V9.4 / LINUX
*   PURPOSE                     : Program to Perform Marginal Structural Model (MSM) Analysis - 2 way.
*   MACROS CALLED               :
*   INPUT                       :
*   OUTPUT                      :
*   PARAMETERS (Optional)       :
*                               :
*   NOTES (Optional)            : ASA Traveling Course, September to October 2025
*
**************************************************************************************************;
**************************************************************************************************/

options noquotelenmax;

** define relevant directories ;
libname ads  '/bdm/tbos/AP24534/studies/201/csr/adhoc/ShortCourse/Clean_SAS_Programs' access=readonly;
%include "/bdm/tbos/AP24534/studies/201/csr/adhoc/ShortCourse/Demo/macro_adj_pvalue.sas";
%include "/bdm/tbos/AP24534/studies/201/csr/adhoc/ShortCourse/Demo/macro_km_plot.sas";


data final;
      set ads.final_casestudy1;
      by usubjid analdy;
      month2=sqrt(month);
      hgb2=sqrt(hgb);
      psa2=sqrt(psa);
      if last.usubjid then analdy=survtime;
run;

***********************************************;
*** creating weight for MSM;
***********************************************;
/* Model 1, weight for taking alternativ therapy, numerator */
ods select ParameterEstimates OddsRatios;
title "Logistic Model 1, weigh for taking alternativ therapy, numerator";
```

```
proc logistic data=final;
      where analdy<=althdy or althdy=.;
      class trt region pdbase pandrog racecat
         alkcat ldhcat visceral acthfl antinefl stra_factor hgbcat;
        model altrx = trt region pandrog racecat ecog month month2
              alkcat ldhcat visceral acthfl  antinefl;
        output out=model1 p=altv_0;
run;


/* Model 2, weight for taking alternativ therapy, denominator */
ods select ParameterEstimates OddsRatios;
title "Logistic Model 2, weigh for taking alternativ therapy, denominator";
proc logistic data=final;
      where analdy<=althdy or althdy=.;
      class trt region pdbase pandrog racecat
           alkcat ldhcat visceral acthfl antinefl stra_factor hgbcat;
        model altrx =trt region pandrog racecat ecog  pd  highpsa month ecogt psa hgb hgb2 psa2 month2
         alkcat ldhcat visceral acthfl  antinefl highpsa;
      output out=model2 p=altv_w;
run;


/* Model 3 , weight for non-informative censoring, numerator */
ods select ParameterEstimates OddsRatios;
title "Logistic Model 3 , weight for non-informative censoring, numerator";
proc logistic data=final;
      class trt region pdbase pandrog racecat
       alkcat ldhcat visceral acthfl antinefl stra_factor hgbcat;
        model censor = trt altrx region pandrog ecog racecat month month2
              alkcat ldhcat visceral acthfl  antinefl;
       output out=model3 p=punc_0;
run;



/* Model 4  weight for non-informative censoring, denominator */
ods select ParameterEstimates OddsRatios;
title "Logistic Model 4  weight for non-informative censoring, denominator";
proc logistic data=final;
      class trt region pdbase pandrog racecat
          alkcat ldhcat visceral acthfl antinefl stra_factor hgbcat;
```

```
        model censor = trt altrx region pandrog racecat ecog  pd  highpsa month ecogt psa hgb hgb2 psa2 month2
               alkcat ldhcat visceral  acthfl  antinefl highpsa;
       output out=model4 p=punc_w;
run;


/* Calculate weight */
proc sort data=model1;
        by usubjid analdy die;
        run;

proc sort data=model2;
        by usubjid analdy die;
        run;

proc sort data=model3;
        by usubjid analdy die;
        run;

proc sort data=model4;
        by usubjid analdy die;
        run;

data main_w;
        merge model1 model2 model3 model4;
        by usubjid analdy die;
/* variables ending with _0 refer to the numerator of the weights
   Variables ending with _w refer to the denominator of the weights */

*** compute stablized and non-stablized weights, stablized weight will be used;
/* reset the variables for a new patient */
        if first.usubjid then do;
       k1_0=1; k2_0=1; k1_w=1; k2_w=1;
       end;
        retain k1_0 k2_0 k1_w k2_w;

/* Inverse probability of censoring weights */
        k2_0=k2_0*punc_0;
        k2_w=k2_w*punc_w;

/* Inverse probability of treatment weights */
       /* patients not on alternative therapy */
```

```
        if analdy<althdy or althdy=. then do;
                k1_0=k1_0*altv_0;
                         k1_w=k1_w*altv_w;
                        end;
       /* patients that start alternative therapy in the current month */
                else if analdy=althdy then do;
                k1_0=k1_0*(1-altv_0);
                         k1_w=k1_w*(1-altv_w);
                        end;
       /* patients that have already started alternative therapy */
                else do;
                         k1_0=k1_0;
                         k1_w=k1_w;
                       end;

/* Stabilized and non stabilized weights */
        stabw=(k1_0*k2_0)/(k1_w*k2_w);
        nstabw=1/(k1_w*k2_w);

       if stabw=. then stabw=1;
run;


*** weight model check;
proc sort data=main_w out=main_w_chk; by trt;
run;

proc univariate data=main_w_chk noprint;
       var stabw;
       by trt;
       output out=chk_trt n=n mean=mean std=std min=min max=max p99=p99 p1=p1 sum=sum;
run;

proc univariate data=main_w_chk noprint;
       var stabw;
       output out=chk_all n=n mean=mean std=std min=min max=max p99=p99 p1=p1 sum=sum;
run;

ods exclude none;
proc print data=chk_all;
      title "check weight: mean should be near 1, MSM weight";
      var n mean std min max;
```

```
run;

**** max is too big, No need to rescale stabw for huge ones;
data main_w1;
      set main_w;
      by usubjid analdy;
      mid=1;
run;

proc sort data=main_w1;
      by trt mid;
      run;

data chk_trt;
      set chk_trt;
      by trt;
      mid=1;
      keep trt p99 mid;
run;

data main_w1a;
      merge main_w1 chk_trt;
      by trt mid;
       q = floor(month/3);
       if stabw>p99 then stabw=min(p99*(1+ranuni(analdy)), stabw);
run;

data wtplot;
      set main_w1a;
       if stabw ne 0 then logwt=log(stabw);
       keep usubjid visit q trt logwt stabw;
run;

proc sort data=wtplot;
      by q trt;
run;

ODS GRAPHICS ON;
proc boxplot data=wtplot;
      title  'After rescaling, check weight: mean should be near 1, MSM weight';
   plot logwt*q;
run;
```

```
ods graphics off;

proc univariate data=main_w1a noprint;
       var stabw;
       output out=chk_w n=n mean=mean std=std min=min max=max p99=p99 p1=p1 sum=sum;
run;

proc print data=chk_w;
      title  'After rescaling, check weight: mean should be near 1, MSM weight';
      var n mean std min max;
run;

proc sort data=main_w1a out=main_w;
      by usubjid analdy;
run;

*** reorganize data for proc phreg;
data main_wb;
      set main_w;
      by usubjid analdy;
*** data error;
       a=lag(analdy);
       if first.usubjid then do;
              start=0;
              stop=analdy;
       end;
             else do;
                    start=a;
                    stop=analdy;
             end;

       if stop=start then stop=stop+0.001;
       if stabw=. then stabw=1;
run;

*** final ITT model;
data main_itt;
      set final;
      by usubjid analdy;
       if last.usubjid;
       stop=analdy;
run;
```

```
ODS GRAPHICS ON;
ods select SurvivalPlot;
proc lifetest data=main_itt notable;
       time stop*die(0);
       strata trt;
      title "Log-rank for OS, reproduce ELM-PC4 result";
      title2 "Using proc lifetest, create data for R OS plot";
run;
ods graphics off;

ods select ModelANOVA;
*** final MSM model, trt*altrx not significant, with p>0.3;
proc phreg data=main_wb covs(aggregate);
        class  trt(ref='0') region pdbase;
         model (start, stop)*die(0)= trt altrx trt*altrx/rl;
        id usubjid;
             strata region pdbase;
             weight stabw;
             title "MSM Cox model for OS, Check signaficance of R by A(t) interaction term";
             title2 "Using proc phreg, counting process data format for multiple records";
run;

ods select ModelANOVA ParameterEstimates;
ods output ParameterEstimates=stat_msm;
proc phreg data=main_wb covs(aggregate);
        class  trt(ref='0') region pdbase;
         model (start, stop)*die(0)= trt altrx/rl;
        id usubjid;
             strata region pdbase;
             weight stabw;
             title "MSM Cox model for OS";
             title2 "Using proc phreg, counting process data format for multiple records";
             ods output ParameterEstimates=msm_hr(where=(parameter='trt') keep=Parameter HazardRatio HRLowerCL
HRUpperCL);
run;

ods select none;
proc phreg data=main_wb covs(aggregate);
        class  trt(ref='0') region pdbase;
         model (start, stop)*die(0)= trt altrx trtt altt/rl;
        id usubjid;
```

```
            strata region pdbase;
            trtt = trt*log(stop);
            altt = altrx*log(stop);
            proportionality_test: test trtt, altt;
            weight stabw;
            title "MSM Cox model for OS";
            title2 "Using proc phreg, counting process data format for multiple records";
            title3 "Look at proportionality test for overall violation of the assumption";
            title4 "Time dependent covariate significant -> variable violate the assumption";
run;
ods exclude all;


***************************************************************************;
*** Call macro to calcuate z-score and p-value from weighted Log-rank test
***************************************************************************;
%macro_adj_pvalue(inds=main_wb        /* Final weighted analysis data */
                  ,method=MSM         /* Weighted method */
                  );

***************************************************************************;
*** Call macro to plot adjusted KM curves
***************************************************************************;
%macro_km_plot(inmain=main_wb              /* Final weighted analysis data */
            ,instratds=stat_msm            /* Final statistic data contains stratified p-value, HR and 95% CI */
            ,inunstratds=stat_uspv_msm     /* Final statistic data contains unstratified p-value */
            ,method=MSM                    /* Weighted method */
            );
```